CICERO Report 2004:01

A description of the Dynamic analysis of Economics of Environmental Policy (DEEP) model

Steffen Kallbekken

April 2004

CICERO
Center for International Climate
and Environmental Research
P.O. Box 1129 Blindern
N-0318 Oslo, Norway
Phone: +47 22 85 87 50
Fax: +47 22 85 87 51
E-mail: admin@cicero.uio.no
Web: www.cicero.uio.no

CICERO Senter for klimaforskning
P.B. 1129 Blindern, 0318 Oslo
Telefon: 22 85 87 50
Faks: 22 85 87 51
E-post: admin@cicero.uio.no
Nett: www.cicero.uio.no

Tittel: Beskrivelse av DEEP modellen

Title: A description of the Dynamic analysis of Economics of Environmental Policy (DEEP) model

Forfatter: Steffen Kallbekken

Author: Steffen Kallbekken

CICERO Report 2004:01
18 sider

CICERO Report 2004:01
18 pages

Finansieringskilde: Norges forskningsråd

Financed by: The Research Council of Norway

Prosjekt: SIP KS: Tverrfaglig modellering av klimapolitikkens konsekvenser

Project: SIP KS: Interdisciplinary modelling of the consequences of climate policy

Prosjektleder: Kristin Aunan

Project manager: Kristin Aunan

Kvalitetsansvarlig: Hege Westskog og Asbjørn Aaheim

Quality manager: Hege Westskog and Asbjørn Aaheim

Nøkkelord: Modellering, klimapolitikk, numeriske modeller

Keywords: Modelling, climate policy, numerical models

Sammendrag: DEEP-modellen er en generell likevektsmodell med flere sektorer, regioner og gasser. Den ble utviklet for å analysere økonomiske aspekter av klimapolitikk, og for å generere utlipsscenarier som kan brukes i andre modeller.

Formålet med rapporten er å gi en beskrivelse av modellen, samt å gjøre programmeringskoden tilgjengelig, slik at det er mulig å forstå og vurdere kritisk resultater fra DEEP.

Abstract: The DEEP model is a multi-sectoral, multi-regional, multi-gas dynamic computable general equilibrium model. It was developed for analysing the economics of climate change policy, and to generate emission scenarios for input to other models.

The purpose of this report is to provide a description of the model, and make the MPSGE programming code available, in order that it be possible to understand and critically examine the results obtained using DEEP.

Språk: Engelsk

Language of report: English

Rapporten kan bestilles fra:
CICERO Senter for klimaforskning
P.B. 1129 Blindern
0318 Oslo

The report may be ordered from:
CICERO (Center for International Climate and Environmental Research – Oslo)
PO Box 1129 Blindern
0318 Oslo, NORWAY

Eller lastes ned fra:
http://www.cicero.uio.no

Or be downloaded from:
http://www.cicero.uio.no

Contents

Acknowledgements

The development of the DEEP was financed by the Norwegian Research Council through the CICERO SIP (Strategic Institute Program) “Interdisciplinary modelling of the consequences of climate policy”. A model development group has been responsible for making the key decisions concerning the model. This group has, with members contributing at various times, consisted of Hege Westskog, Asbjørn Aaheim, Cathrine Hagem, Ottar Mæstad, Hans Kolshus and Steffen Kallbekken. Steffen Kallbekken has been responsible for the programming and testing of the model.

1 Introduction

The DEEP model is a multi-sectoral, multi-regional, multi-gas dynamic computable general equilibrium model. It was developed for analysing the economics of climate change policy, and to generate emission scenarios for other models. So far it has primarily been used to analyse issues relating to the Kyoto Protocol, such as enforcement and the Clean Development Mechanism. It is also intended for use in analysing future (post-Kyoto) climate agreements, and for integration with other models (atmospheric chemistry and general circulation models).

Chapter 2 explains what data sources we have used, and how we have manipulated these data. Chapter 3 then outlines the flow of goods, and the general structure of the model. The fourth chapter describes in greater detail the structure of production, demand and trade in the DEEP model. Chapter 5 deals with the treatment of greenhouse gas emissions in the model, while chapter 6 explains the dynamics of the model. Finally, the appendix provides the MPSGE code for the model.

2 The data

The Global Trade Analysis Project (GTAP) is a global data base that contains “bilateral trade, transport, and protection data characterizing economic linkages among regions, together with individual-country input-output data bases that account for intersectoral linkages within each region” (Hertel 1997). Version 5 of the data base (Center for Global Trade Analysis, 2001) consists of data for 57 sectors and 66 regions. The original data base is converted into a GAMS readable format using the GTAPinGAMS conversion tool (Rutherford and Paltsev 2000). These data are then converted to all the parameters required to describe a full social accounting matrix (SAM). A SAM is a general and consistent macroeconomic accounting framework. In this case it is a global SAM consisting of input-output matrices for each region (inputs of primary factors to sectors, and output of consumption and investment), and with trade between the regions. The SAM for the DEEP model is constructed in the same way as in the dynamic GTAP-EG model1.

The DEEP model allows some flexibility in the aggregation of sectors and regions according to the needs of the project in question. There is full flexibility regarding the aggregation of regions. Certain sectors, however, have to be present in order for the model to work. These are the energy sectors crude oil, petroleum, coal, gas and electricity, and agriculture. While the GTAP data base has five primary factors (capital, land, natural resources, skilled and unskilled labour), this is aggregated to just two primary factors in DEEP – capital (capital, land and natural resources) and labour (skilled and unskilled labour). The GTAP regions and sectors are listed in appendix 2.

Emissions data for CO_2, N_2O and CH_4 are provided by the GTAP/EPA project “Towards An Integrated Data Base for Assessing the Potential for Greenhouse Gas Mitigation” (Lee 2002 and Lee 2003). CO_2 emissions are provided for the use of six energy commodities, as intermediate inputs to all 57 GTAP sectors, for each GTAP region, and divided between imported and domestic sources (a 6*57*57*66*2 matrix). These data we aggregate to a three commodity (oil, coal and gas) emission data base for intermediate inputs in each region, without separation of domestic and imported sources (a 3*57*57*66 matrix). Emissions data for nitrous oxide (N_2O) and methane (CH_4) are from the US Environmental Protection Agency. Emissions are provided with a sectoral breakdown.

[1] With the exception that we do not separate out the capital used in fossil fuel production as resource capital.

The whole emissions data base is stored in the model, but for each run of the model, emissions are calculated with the appropriate sectoral and regional breakdown, using an aggregation procedure (aggrghg.gms). This ensures that the emissions data do not limit the flexibility in regional and sectoral aggregation.

3 About the model

The DEEP model builds on a combination of other models, but also differs from each of these in significant respects. The structure of production and demand is based on the GTAP-EG model by Rutherford and Paltsev (2000), with some modifications (primarily to the structure of final demand). The dynamics of the model were developed with great help from the "notes" by Paltsev (1999). Non-CO_2 emissions are implemented using the approach adopted in the EPPA model (Hyman et al. 2002).

The model is programmed in the MPSGE (Mathematical Programming System for General Equilibrium Analysis). MPSGE is a programming language that was developed by Tom Rutherford for solving Arrow-Debreu economic equilibrium models.

The DEEP model consists of five main elements: Production, a representative agent (demanding a consumption good, receiving tax revenue, and having primary factor endowments), an Armington aggregation of domestic and imported goods, and a capital and an investment sector. The model structure is presented in figure 1. For simplicity, emissions trading has been excluded from this figure. Parameter names are given as the different elements of the models are presented.

In the basic model there are eight sectors and nine regions, which are listed in table 1. The production of goods (**py**) in each sector (**y**) takes place using the primary factors (**pl** and **rk**) and inputs from other sectors (domestic - **py** or imported - **pa**). With the exception of the primary factors and the investment good (**py(cgd)**), all goods can be traded bilaterally between all regions. Tariffs on imports and exports, and taxes on inputs and outputs are included in the model (in terms of rates reported by GTAP). All trading takes place through an Armington aggregation (**a**) of domestic and imported goods. Transport margins (**pt**), a fixed-share cost of transportation, are applied to all such international (bilateral) trades – giving rise to a price differential between world prices for exports and imports (transportation applied to exports before import price is calculated). The transport margin is produced by an international transport sector (yt) with inputs from the production sectors (**py**). The consumption good (**pc**) is made up of inputs from all sectors (after domestic and imported goods have been aggregated to an Armington composite). Thus there is not direct consumption of imported goods. This consumption good is the only good that the representative agent (**ra**) demands.

The representative agent is endowed with the two primary factors, labour (**pl**) and capital (**pk**), and with emission permits. The sale of these endowments generates income for the representative agent. All revenues from taxes and tariffs also accrue to the agent (which we recall is both consumer and government).

The capital which the representative agent is endowed with (**pk**), is purchased by a capital sector (**kap**). This sector rents out capital to the production sectors (receiving the interest (**rk**) in return). The capital stock minus the per-period depreciation is carried forward to the next time period. The capital stock can grow through investments. The investment sector (**inv**) produces capital (**pk**) by purchasing a capital (investment) good (**py(cgd)**) from the non-fossil fuel production sector.

All sectors that emit carbon dioxide (CO_2), methane (CH_4) or nitrous oxide (N_2O), in regions that are part of a climate agreement, are required to hold permits corresponding to the amount of emissions. These permits are initially held by the representative agent in each region, who are endowed with a certain amount of permits –corresponding to their emission limits. Because the national target is for a certain amount of greenhouse gas (GHG) emissions (measured in CO_2-equivalents), this endowment is given in the form of general GHG permits. A sector converts the general GHG permits to the specific emission permits for each gas – that the emitters are required to hold. These specific-gas permits are not region specific, and can be traded among all regions that are part of the climate agreement (unlike for other goods, where all trades are bilateral). Because of the complexity of the flow of goods, emission permits are not included in figure 1.

Table 1 Regions and sectors in DEEP

Regions		**Sectors**	
EUU	European Union	**Non-fossil fuel production**	
JPN	Japan	AGR	Agriculture
FSU	Former Soviet Union countries	EIS	Energy intensive goods
CEE	Central and Eastern Europe	ELE	Electricity
RAB	Rest of Annex B countries	Y	Other manufactures and services
KRC	Kyoto-rejecting countries (USA and Australia)	CGD	capital (investment) good
CHN	China	OIL	Refined oil
OPE	OPEC countries		
ROW	Rest of world	**Fossil fuel production**	
		CRU	Crude oil
		GAS	Natural gas
		COL	Coal



Figure 1 **Flow of goods in the DEEP model (excluding emission permits)**

4 Structure of production sectors and final demand

The structure of production and demand has been adopted from the GTAP-EG model by Rutherford and Paltsev (2000). This chapter therefore follows closely their description of the model. All unreferenced quotes in this chapter are Rutherford and Paltsev (2000).

The structure of DEEP deviates from the GTAP-EG model in three respects. The first is the inclusion of methane and nitrous oxide. How emissions and emission permits (including CO_2) are implemented in the model, will be described in the next chapter. The second is that the nesting structure for final demand has been changed to reflect what we believe are more realistic assumptions concerning energy use. Because the DEEP model is dynamic, while the GTAP-EG model is static, all sectors and goods have a time dimension. Also, in some cases technology improvement, or growth variables have been introduced. These are discussed in chapter 7 (dynamics of the model).

4.1 Production

Production is described using two different production functions, one for fossil fuel production, and one for non-fossil fuel production. Figure 2 shows the production structure for fossil fuel production. The values to the right of each node in the tree show the elasticity. The name of the nodes refers to the MPSGE formulation listed in appendix 1.

Fossil fuel production is a CES-function that includes crude oil, gas and coal. The output is produced as an aggregate of a resource (**rk**) and a non-resource input. The non-resource input is a Leontief composite of labour (**pl**) and the Armington composite good (**pa**). The elasticity of substitution between the resource and non-resource inputs (**ese**) depends on the (fixed) value share of resource inputs in fossil fuel supply.



Figure 2 Fossil fuel production

Figure 3 describes the nesting structure and elasticities for non-fossil fuel production. “Output is produced with fixed-coefficient (Leontief) inputs of intermediate non-energy goods and an energy-primary factor composite (**vae**). The energy-primary factor composite is a constant-elasticity of substitution (CES) function with an elasticity of 0.5. The composite is in turn made from an energy composite and a primary factor composite (**va**) of labor (**pl**) and capital (**rk**). The primary factor composite is aggregated through a Cobb-Douglas production function. The energy composite is a CES function of electricity (**pa(ele)**) versus a non-electricity (**nel**) energy composite, with an elasticity of 0.1. The energy composite is made up of coal (**pa(col)**) and a liquid fuel (**lqd**) composite, with an elasticity of 0.5. At the very bottom of the nest, we find this liquid fuel composite of oil (**pa(oil)**) and gas (**pa(gas)**) with an elasticity of substitution of 2.



Figure 3 Non-fossil fuel production

4.2 International trade

In DEEP the world consists of different regional economies that are linked though bilateral trade flows. All goods, except the primary factors (labour and capital) and the investment good, can be traded among the regions. The model assumes that goods produced in different countries are not identical (the "Armington assumption"). Trades therefore take place between pairs of countries, rather than importing and exporting through some global pool of the good. Each bilateral trade flow requires its own transportation service (with the exception of emission permits).

Figure 4 shows the structure of the Armington aggregation of domestic (**py(i,r)**) and imported (**py(i,s)**) goods. The elasticity of substitution between domestic and imported goods (**s**) is 4, while the, while the elasticity of substitution among imports from different regions (**m**) is 8. When a region imports a good, it is required to pay for transportation. This is modelled as a Leontief technology between the imported good and the transportation good (**pt**).

The transportation margins are "proportional to quantities traded". The international transport services themselves are "assumed to be a Cobb-Douglas composite of goods provided in the domestic market in each region."



Figure 4 Armington supply

4.3 Final demand/consumption

The structure of final demand is shown in figure 5. Final demand consists of one good, the consumption good **c(r)**, which is a "constant elasticity aggregate of non-energy consumption and energy. The non-energy composite (**c**) is in turn a Cobb-Douglas aggregate of different goods."

The structure of the energy composite (**e**) is different in DEEP from GTAP-EG. In DEEP it is an aggregate of oil (**pa(oil)**) and a composite (**oe**) of electricity, petroleum and gas, with an elasticity of 0.3. The oe composite is in turn an aggregate of fossil energy (**of**) and electricity (**pa(ele)**) with an elasticity of 0.75,. The fossil energy composite is an aggregate of coal (**pa(col)**) and gas (**pa(gas)**) with an elasticity of substitution of 0.6.

The reasoning behind the design of this structure for energy demand is the following; the first node (**e**) is a choice between mobile (transportation) and stationary energy (primarily power generation). Transportation is represented by oil (96% of all energy use in transportation is oil, and 49% of all oil goes to transportation), while power generation consists of electricity, gas and coal. 100% of coal and 99% of gas goes into non-transport energy use. Stationary energy use is then, in turn, a choice between electricity and fossil fuels (coal and gas).



Figure 5 Nesting structure for consumption good

5 Representative agent

The representative agent, which is both consumer and government, demands only the consumption good. To pay for this good, the agent is endowed with labour and capital. The labour endowment grows for each year – at the same rate as the growth parameter (see chapter 7), while the capital endowment is only given in the first year (an initial capital stock which has to be increased through investment). The representative agent collects all taxes and tariffs specified in the model. If the region takes part in a climate agreement, then the agent is also endowed with emission permits.

The agent uses the tax revenue and income from endowments to purchase the consumption good – or pay for investment. While the agent gets utility only from the consumption good, investment is driven by the returns to capital generated in the next period (which will of course increase consumption), and a negative endowment of terminal capital. The function of this negative endowment is discussed in chapter 7.

6 Greenhouse gas emissions

In the real world CO_2 is emitted in relatively constant proportions to the use of fossil fuels. This makes CO_2 emissions relatively easy to model – particularly when good and detailed emissions data exist: For each amount of fossil fuels used in final demand, or in intermediate demand, a certain amount of CO_2 is emitted (the GTAP/EPA data tell us exactly how much). When there are restrictions on greenhouse gas (GHG) emissions, for example if we are looking at implementation of the Kyoto Protocol, each producer emitting GHGs has to hold a quantity permits corresponding to the amount of gases emitted. In the model this is done by requiring a fixed share input of permits (more technically, through a Leontief technology composite of fossil fuels and permits). This is the same approach as in the GTAP-EG model.

Non-CO_2 emissions are modelled adopting the approach of the EPPA model (Hyman et al. 2002). Emission permits for CH_4 and N_2O are modelled as direct inputs to the production function (where there are emissions) at the top of the nest. In doing this we follow Hyman et al. (2002). “This formulation implies, other things equal, that a rise in the price of the GHG (as a result of tighter emissions constraint) leads to a substitution away from it by means of a proportional increase in the use of all other inputs. […] locating GHG disposal at the top of the nest, and the implications of doing so, represent in our judgment a good first approximation of input demand resulting from emissions control" (Hyman et al. 2002).

To be more precise, a sector that emits CH_4 and or N_2O produces a good that at the top of the nesting structure is a composite of CH_4 and a N_2O-good composite. This second composite is, as the term would imply, a composite of N_2O and the actual good. The elasticities are taken from Hyman et al. They derive estimates for the elasticity of substitution for CH_4 and N_2O for each sector (and with a further regional differentiation for agricultural emissions). They do this based on estimates of marginal abatement cost curves. Where the regional aggregation is not the same in DEEP as in Hyman et al., the elasticities for agriculture have to be averaged through a weighting procedure.

7 Dynamics of the model

The model is an intertemporal model (developed according to the guidelines of Paltsev 1999) with a utility maximising representative agent. Investment (growth) is endogenous, but investment is not determined, as in many other models, through a time preference rate or savings rate. Instead we have a growth parameter that defines a growth rate that is optimal for the original equilibrium, i.e. when there are no shocks (for example before a climate policy is introduced). The growth rate can be different between regions and time periods.

While we do not have an explicit time preference, the growth parameter does of course imply a specific time preference rate - as achieving a specific level of growth requires a specific level of investments (or a specific consumption-investment trade-off). What the growth parameter does is to tell the model what growth rates will maximise utility as long as everything remains the same. Choosing the growth parameter is therefore a matter of developing a growth scenario that we believe can take place – and that reflects utility maximisation.

When we introduce a shock to the equilibrium, the agent will maximise utility by adapting in various ways; substituting between sectors and inputs – and also between time periods. This intertemporal substitution can reduce or increase investments in a specific time period compared to the pre-defined growth parameters. The resulting economic growth might therefore differ from the growth parameter that defines the optimum for the no-shock situation.

The dynamics of the model are determined through setting certain parameters and through three blocks in the MPSGE formulation. The parameters are the growth parameter (annual growth), interest rate, depreciation rate, number of time periods, and years in each time period. The dynamic parameters can be changed between scenarios. In our standard scenario, the growth rate follows the IPCC SRES scenario 1B, while the interest and depreciation rates are both set to 5% per year. The three MPSGE blocks are the two production blocks capital and investment, and the demand block (representative agent).

The capital sector (**cap**) takes capital stock (**pk(t)**) as its input, while the output is return to capital (**rk**) – and next-period capital stock (**pk(t+1)**). The return to capital is determined by the interest rate, while the next-period capital stock is equal to the initial capital stock less depreciation. The GTAP data does not provide us with information about what the initial capital stock is, only about the value of return to capital. The size of the capital stock is therefore inferred from the rate of return (capital earnings equal capital stock multiplied by the rental price). The capital stock can increase through investments.

Investment (**inv**) takes place through the production of an investment good (**py(cgd)**). The investment good is produced in the same way as other non-fossil fuel goods. The output from the investment is capital stock – in the next time period. The GTAP data on investments will not be consistent with the investment rates produced in the model (unless they should happen to exactly match the rates implied by the GTAP data). One option would be to calibrate the interest rate so that the observed (GTAP) and the predicted investment rates (endogenous to the model) were in line. We have instead adopted the approach used in the Dynamic GTAP model (Paltsev 1999), where the observed investment in the GTAP data base is discarded, and the SAM is recalibrated to a new level of investment – consistent with the growth parameter. To achieve this, the investment rule is:

Investment = capital stock * (growth rate + interest rate)

This calibration of investment has to be done if we want to run the model for any growth rates other than those implied by the investment choices observed by GTAP.

The representative agent has an endowment of labour that grows (exogenously) parallel to the initial growth rates (the growth parameter). The representative agent is also given a negative endowment of final-period capital. This is done in order to avoid zero investments in the final period; if there was no such endowment, the representative agent would consume all capital in the final period, and make no investments. While this is rational for a finite model horizon – as the agent will never see the returns from any final period investments, we want to model a situation where the economy goes on beyond the horizon of the model. We therefore has to force the agent to make investments in the final period. One way of doing this is to require the agent to hold a certain amount of capital stock by the final period. The exact quantity is set endogenously such that the ratio of investments to consumption is the final period is the same as in the period before that.

References

Center for Global Trade Analysis, 2001, *The GTAP5 Data Package – Global Trade, Assistance and Production*, CD-ROM, Purdue University.

Hertel T. (ed.), 1997, *Global Trade Analysis – Modeling and Applications*, Cambridge University Press, Cambridge, p.4.

Hyman R.C., Reilly J.M., Babiker M.H., De Masin, Jacoby H.D., 2002, Modeling Non-CO2 Greenhouse Gas Abatement, MIT Joint Program on the Science and Policy of Global Change, *Report No. 94*.

Lee H.L., 2002, An Emissions Data Base for Integrated Assessment of Climate Change Policy Using GTAP, *GTAP Working Paper (draft)*, Center for Global Trade Analysis, Purdue University.

Lee H.L., 2003, The GTAP Non-CO2 Emissions Data Base*, GTAP Resource #1186*, http://www.gtap.agecon.purdue.edu/resources/res_display.asp?RecordID=1186.

Paltsev S.V., 1999, *Moving from Static to Dynamic General Equilibrium Economic Models (Notes for a beginner in MPSGE)*, Joint Program on the Science and Policy of Global Change, Massachusetts Institute of Technology, http://debreu.colorado.edu/papers/newmove.pdf.

Rutherford T.F., Paltsev S.V., 2000, GTAP-Energy in GAMS: The Dataset and Static Model, *Working Paper No. 00-2*, Department of Economics, University of Colorado.

Appendix 1: MPSGE code for DEEP

```
$sectors:
        c(r,t)                  ! Private consumption
        y(i,r,t)$vom(i,r)       ! Output
        a(i,r,t)$a0(i,r)        ! Armington aggregation
        yt(t)                   ! Transport
        conv(ghg,r)$pr(r)       ! Conversion from endowment to individual ghg permits
        kap(r,t)                ! Capital accumulation
        inv(r,t)                ! Investment

$commodities:
        pc(r,t)                 ! Consumption good
        py(i,r,t)$vom(i,r)      ! Output
        pa(i,r,t)$a0(i,r)       ! Armington composite
        pl(r,t)                 ! Wage rate
        rk(r,t)                 ! Return to capital
        pk(r,t)                 ! Capital
        pkt(r)                  ! Terminal capital
        pt(t)                   ! Transport services
        pp(ghg,r)$pr(r)         ! Individual ghg permit
        ppe(r)$pr(r)            ! ghg endowment permit

$consumers:
        ra(r)                   ! Representative agent


$auxiliary:
        tk(r)                   ! Terminal capital stock
        theta(r)                ! Steady-state share of assets
        ta(r)                   ! Terminal asset adjustment

*       Final demand

$prod:c(r,t)            s:sub(r) h(s):sbs(r)        g(h):0.5 c(g):1    e(g):0.3
```

```
+       oil(e):0          oe(e):0.75        of(oe):0.6        gas(of):0         col(of):0
        o:pc(r,t)         q:ct0(r)
        i:pa(i,r,t)       q:c0(i,r)p:pc0(i,r)                 a:ra(r)   t:tc(i,r)  c:$(not e(i))
+       oil:$oil(i)       oe:$ele(i)        gas:$gas(i)       col:$col(i)
        i:pp(co2,r)#(oil)$m(r,t)            q:(co2s(oil,r)*imp(r,t))            oil:
        i:pp(co2,r)#(col)$m(r,t)            q:(co2s(col,r)*imp(r,t))            col:
        i:pp(co2,r)#(gas)$m(r,t)            q:(co2s(gas,r)*imp(r,t))            gas:
        i:pp(ch4,r)$m(r,t)                  q:(ch4f(r)*imp(r,t))
        i:pp(n2o,r)$m(r,t)                  q:(n2of(r)*imp(r,t))                h:

*     Non-fossil fuel production

$prod:y(i,r,t)$nr(i,r)        s:(sch4(i,r))     h(s):(sn2o(i,r))  g(h):0    vae(g):0.5
+       va(vae):1             e(vae):0.1        nel(e):0.5        lqd(nel):2
+       oil(lqd):0            col(nel):0        gas(lqd):0
        o:py(i,r,t)           q:vom(i,r)        a:ra(r)           t:ty(i,r)
        i:pa(j,r,t)$(not fe(j))  q:vafm(j,i,r)  p:pai0(j,i,r)     a:ra(r)   t:ti(j,i,r)
+       g:$(not erg(j)) e:$ele(j)
        i:pl(r,t)             q:ld0(i,r)        va:
        i:rk(r,t)             q:kd0(i,r)        va:
        i:pa(fe,r,t)          q:vafm(fe,i,r)    p:pai0(fe,i,r)    fe.tl:    a:ra(r)   t:ti(fe,i,r)
        i:pp(co2,r)#(fe)$m(r,t)  q:(fi(fe,i,r)*imp(r,t))          fe.tl:
        i:pp(ch4,r)$m(r,t)    q:(ch4s(i,r)*imp(r,t))
        i:pp(n2o,r)$m(r,t)    q:(n2os(i,r)*imp(r,t))              h:


*     Fossil fuel production activity (crude, gas and coal)

$prod:y(xe,r,t)$vom(xe,r)     s:(sch4(xe,r))    h(s):(sn2o(xe,r))           c(h):0
+       g(c):(ese(xe,r))      id(g):0
        o:py(xe,r,t)          q:vom(xe,r)       a:ra(r)           t:ty(xe,r)
        i:pa(j,r,t)           q:vafm(j,xe,r)    p:pai0(j,xe,r)    a:ra(r)   t:ti(j,xe,r)        id:
        i:pl(r,t)             q:ld0(xe,r)       id:
        i:rk(r,t)             q:kd0(xe,r)       g:
        i:pp(co2,r)$m(r,t)    q:(sum(fe, fi(fe,xe,r))*imp(r,t))  c:
        i:pp(ch4,r)$m(r,t)    q:(ch4s(xe,r)*imp(r,t))
```

```
        i:pp(n2o,r)$m(r,t)     q:(n2os(xe,r)*imp(r,t))     h:

*   Armington aggregation

$prod:a(i,r,t)$a0(i,r)   s:4    m:8    s.tl(m):0
        o:pa(i,r,t)     q:(eff(i,r,t)*a0(i,r))
        i:py(i,r,t)     q:(gdp(r,t)*d0(i,r))
        i:py(i,s,t)     q:(gdp(s,t)*vxmd(i,s,r))   p:pmx0(i,s,r)   a:ra(s)   t:tx(i,s,r)
+       a:ra(r)         t:(tm(i,s,r)*(1+tx(i,s,r)))   s.tl:
        i:pt(t)#(s)     q:(gdp(s,t)*vtwr(i,s,r))   p:pmt0(i,s,r)   a:ra(r)   t:tm(i,s,r)     s.tl:


*   International transport

$prod:yt(t)             s:1
        o:pt(t)         q:(sum((i,r), vst(i,r)))
        i:py(i,r,t)     q:vst(i,r)


*       Conversion of GHG permits

$prod:conv(ghg,r)$pr(r)
        o:pp(ghg,r)             q:permqt(ghg,r)
        i:ppe(r)$pr(r)          q:permqt(ghg,r)


*       Capital accumulation

$prod:kap(r,t)
        o:pk(r,t+1)             q:((1-delta)*k0(r))
        o:pkt(r)$tlast(t)       q:((1-delta)*k0(r))
        o:rk(r,t)               q:(k0(r)*(delta+rate))
        i:pk(r,t)               q:k0(r)


*       Investment
```

```
$prod:inv(r,t)
        o:pk(r,t+1)             q:i0(r)
        o:pkt(r)$tlast(t)       q:i0(r)
        i:py(cgd,r,t)           q:i0(r)


*       Final demand:

$demand:ra(r)                   s:0.5
        d:pc(r,t)               q:(ct0(r)*qref(r,t))              p:pref(r,t)
        e:pl(r,t)               q:(evoa("lab",r)*qref(r,t))
        e:pl("krc",tlast)  q:(-1)   r:ta(r)
        e:pk(r,tfirst)          q:k0(r)
        e:pkt(r)                q:(-k0(r))                        r:tk(r)
        e:pc("krc",t)           q:(vb(r)*qref(r,t))
        e:ppe(r)$pr(r)          q:ghglim(r)

*       Terminal capital constraint

$constraint:tk(r)
        sum(t$tlast(t+1), c(r,t)*inv(r,t+1) - c(r,t+1)*inv(r,t)) =e= 0;


*       Steady-state share of assets

$constraint:theta(r)
theta(r) * sum((rr,tlast), c(rr,tlast)*pc(rr,tlast)*ct0(rr) - evoa("lab",rr)*qref(rr,tlast)*pl(rr,tlast)
        - vb(rr)*qref(rr,tlast)*pc("euu",tlast) - ppe(rr)*ghglim(rr))
        =e= sum(tlast, c(r,tlast)*pc(r,tlast)*ct0(r) - evoa("lab",r)*qref(r,tlast)*pl(r,tlast)
        - vb(r)*qref(r,tlast)*pc("euu",tlast) - ppe(r)*ghglim(r));

*       Terminal asset adjustment

$constraint:ta(r)
sum(tfirst, pl("krc",tfirst)) * ta(r) =e= theta(r) * sum(rr,pkt(rr)*tk(rr)*k0(rr)) -
pkt(r)*tk(r)*k0(r);
```

Appendix 2: GTAP regions and sectors

GTAP regions

AUS	Australia
NZL	New Zealand
CHN	China CHN
HKG	Hong Kong
JPN	Japan
KOR	Korea, Republic of KOR
TWN	Taiwan
IDN	Indonesia
MYS	Malaysia
PHL	Philippines
SGP	Singapore
THA	Thailand
VNM	Vietnam
BGD	Bangladesh
IND	India
LKA	Sri Lanka
XSA	Rest of South Asia: Bhutan, Maldives, Nepal, Pakistan
CAN	Canada
USA	United States of America: American Samoa, Guam, Northern Mariana Islands, USA
MEX	Mexico
XCM	Central America and the Caribbean: Anguila, Antigua & Barbuda, Aruba, Bahamas, Barbados, Belize, Cayman Islands, Costa Rica, Cuba, Dominica, Dominican Republic, El Salvador, Grenada, Guatemala, Haiti, Honduras, Jamaica, Netherlands Antilles, Nicaragua, Panama, Saint Kitts and Nevis, Saint Lucia, Saint Vincent and the Grenadines, Trinidad and Tobago, British Virgin Islands
COL	Colombia
PER	Peru
VEN	Venezuela
XAP	Rest of Andean Pact: Bolivia, Ecuador
ARG	Argentina
BRA	Brazil
CHL	Chile
URY	Uruguay
XSM	Rest of South America: Guyana, Paraguay, Suriname

AUT	Austria
BEL	Belgium
DNK	Denmark
FIN	Finland
FRA	France: France, French Guiana, Martinique, Reunion
DEU	Germany
GBR	United Kingdom
GRC	Greece
IRL	Ireland
ITA	Italy
LUX	Luxembourg
NLD	Netherlands
PRT	Portugal
ESP	Spain
SWE	Sweden
CHE	Switzerland
XEF	Rest of EFTA: Iceland, Liechtenstein, Norway
HUN	Hungary
POL	Poland
XCE	Rest of Central European Associates: Bulgaria, Czech Republic, Romania, Slovakia, Slovenia
XSU	Former Soviet Union: Armenia, Azerbaijan, Belarus, Estonia, Georgia, Kazakhstan, Kyrgyzstan, Latvia, Lithuania, Republic of Moldova, Russian Federation, Tajikistan, Turkmenistan, Ukraine, Uzbekistan
TUR	Turkey
XME	Rest of Middle East: Bahrain, Islamic Republic of Iran, Iraq, Israel, Jordan, Kuwait, Lebanon, Oman, Qatar, Saudi Arabia, Syrian Arab Republic, United Arab Emirates, Yemen
MAR	Morocco
XNF	Rest of North Africa: Algeria, Egypt, Libyan Arab Jamahiriya, Tunisia
BWA	Botswana
XSC	Rest of South African Customs Union: Lesotho, Namibia, South Africa, Swaziland
MWI	Malawi
MOZ	Mozambique
TZA	Tanzania
ZMB	Zambia
ZWE	Zimbabwe
XSF	Other Southern Africa: Angola, Mauritius

UGA Uganda

XSS Rest of Sub-Saharan Africa: Benin, Burkina Faso, Burundi, Cameroon, Cape Verde, Central African Republic, Chad, Comoros, Congo, Cote d'Ivoire, Djibouti, Equatorial Guinea, Eritrea, Ethiopia, Gabon, Gambia, Ghana, Guinea, Guinea-Bissau, Kenya, Liberia, Madagascar, Mali, Mauritania, Mayotte, Niger, Nigeria, Rwanda, Sao Tome and Principe, Senegal, Seychelles, Sierra Leone, Somalia, Sudan, Togo, Zaire

XRW Rest of World: Afghanistan, Albania, Andorra, Bermuda, Bosnia and Herzegovina, Brunei, Cambodia, Croatia, Cyprus, Faroe Islands, Fiji, French Polynesia, Gibraltar, Greenland, Guadeloupe, Kiribati, Lao People's Democratic Republic, Macau, the former Yugoslav Republic of Macedonia, Malta, Marshall Islands, Federated States of Micronesia, Monaco, Mongolia, Myanmar, Nauru, New Caledonia, Democratic People's Republic of Korea, Papua New Guinea, San Marino, Solomon Islands, Tonga, Tuvalu, Vanuatu, Western Samoa, Yugoslavia

GTAP sectors

PDR Paddy rice

WHT Wheat

GRO Cereal grains nec

V_F Vegetables, fruit, nuts

OSD Oil seeds

C_B Sugar cane, sugar beet

PFB Plant-based fibers

OCR Crops nec

CTL Bovine cattle, sheep and goats, horses

OAP Animal products nec

RMK Raw milk

WOL Wool, silk-worm cocoons

FOR Forestry

FSH Fishing

COL Coal

OIL Oil

GAS Gas

OMN Minerals nec

CMT Bovine meat products

OMT Meat products nec

VOL Vegetable oils and fats

MIL Dairy products

PCR Processed rice

SGR Sugar

OFD Food products nec

B_T	Beverages and tobacco products
TEX	Textiles
WAP	Wearing apparel
LEA	Leather products
LUM	Wood products
PPP	Paper products, publishing
P_C	Petroleum, coal products
CRP	Chemical, rubber, plastic products
NMM	Mineral products nec
I_S	Ferrous metals
NFM	Metals nec
FMP	Metal products
MVH	Motor vehicles and parts
OTN	Transport equipment nec
ELE	Electronic equipment
OME	Machinery and equipment nec
OMF	Manufactures nec
ELY	Electricity
GDT	Gas manufacture, distribution
WTR	Water
CNS	Construction
TRD	Trade
OTP	Transport nec
WTP	Water transport
ATP	Air transport
CMN	Communication
OFI	Financial services nec
ISR	Insurance
OBS	Business services nec
ROS	Recreational and other services
OSG	Public Administration, Defense, Education, Health
DWE	Dwellings